

# SPICE - Lucrarea 1

## Recapitulare SPICE

### 1. Definirea componentelor pasive:

Rezistoare	R<nume> <N+> <N-> [nume_model] <valoare> [TC=<TC <sub>1</sub> > [,<TC <sub>2</sub> >]]
Condensatoare	C<nume> <N+> <N-> [nume_model] <valoare> [IC=<condiție_inițială>]
Inductanțe	L<nume> <N+> <N-> [nume_model] <valoare> [IC=<condiție_inițială>]
Transformatoare ideale	L<primar> <NP+> <NP-> <valoare_primar> L<secundar> <NS+> <NS-> <valoare_secundar> K<nume> L<primar> L<secundar> <coeficient_cuplaj>

#### **Exemple:**

R102a 1 0 100  
C13 4 7 25uF  
L1 4 3 1H  
L2 2 0 0.01H  
K12 L1 L2 0.99

#### **Observații:**

- 1) .MODEL <nume\_model> <identificator>([parametru\_model=valoare...])  
unde <identificator> poate fi RES, CAP sau respectiv IND.
- 2) Pentru luarea în considerare a valorilor inițiale (tensiune pentru capacități, respectiv curenți pentru inductanțe) specificate cu IC, la analiza de regim tranzitoriu în comanda TRAN se face specificația UIC.


### 2. Definirea surselor independente de tensiune și curent:

Tensiune	V<nume> <N+> <N-> [[DC=]<valoare>] [AC <ampl> [fază]] [specific_regim_tran]
Curent	I<nume> <N+> <N-> [[DC=]<valoare>] [AC <ampl> [fază]] [specific_regim_tran]

Sinusoidal	SIN(<Voff> <Vampl> <frec> <td> <df_amortizare > <fază>) SIN(<Ioff> <Iampl> <frec> <td> <df_amortizare > <fază>)
Pulsatoriu	PULSE(<A1> <A2> <td> <tr> <tf> <pw> <per>)
Liniar pe porțiuni	PWL(t1,A1) (t2,A2) ... (tn,An)
Modulat în frecvență	SFFM(<Voff> <Ampl> <FC> <M> <FM>)
Exponențial	EXP(<A1> <A2> <td1> <tc1> <td2> <tc2>)

#### **Exemple:**

Vcc 4 0 DC 10V  
IE 4 1 DC 100mA  
Vin 1 0 AC 10mV 90  
Igen 0 1 AC 50mA 180  
Vg 2 0 DC=3V AC 10mV 30 SIN(0 1 1kHz 0 0 0)  
Ig 5 1 SIN(0A 100mA 1kHz 0 0 30)  
V1 6 0 PULSE(-3V 5V 10ms 1ms 1ms 10s 20s)  
Vcom 7 0 PWL (0,0) (1ms,1V) (2ms,1V) (3ms,5V) (10ms,5V)



Forma de undă a semnalului de tip PULSE

### 3. Definirea surselor comandate de tensiune și curent:

Comandă Liniară	<p>E&lt;nume&gt; &lt;N+&gt; &lt;N-&gt; &lt;NC+&gt; &lt;NC-&gt; &lt;câștig&gt;            G&lt;nume&gt; &lt;N+&gt; &lt;N-&gt; &lt;NC+&gt; &lt;NC-&gt; &lt;transconductanța&gt;            H&lt;nume&gt; &lt;N+&gt; &lt;N-&gt; &lt;NC+&gt; &lt;NC-&gt; &lt;transrezistența&gt;            F&lt;nume&gt; &lt;N+&gt; &lt;N-&gt; &lt;NC+&gt; &lt;NC-&gt; &lt;câștig&gt;</p>
Susre ABM	<p>E&lt;nume&gt; &lt;N+&gt; &lt;N-&gt; &lt;cuvânt_cheie_ABM&gt;=&lt;funcție_ABM&gt;            G&lt;nume&gt; &lt;N+&gt; &lt;N-&gt; &lt;cuvânt_cheie_ABM&gt;=&lt;funcție_ABM&gt;, unde:            cuvânt_cheie_ABM specifică tipul funcției de transfer și poate fi:                VALUE – expresie matematică                TABLE – tabel                LAPLACE – transformată Laplace                FREQ – tabel cu răspunsul în frecvență                CHEBYSHEV – funcție de transfer Cebășev            funcție_ABM – specifică funcția de transfer în formatul cerut de cuvântul cheie ABM (expresie matematică sau tabel)</p>

#### Exemple:

```

E1 3 0 12 14 15 ;E1=15*(V(14)-V(15))
Gm 5 2 3 12 0.1 ;gm=0.1*(V(3)-V(12))
H1a 8 4 Vcom 2 ;H1a=2*I(Vcom)
F0 3 2 Vcontrol 10 ;F0=10*I(Vcontrol)
Emodulator 7 0 VALUE={V(1)*V(2,3)*GAIN} ;modulator
Gfiltru OUT 0 LAPLACE{V(IN)}={1/(1+s/(6.28*1kHz))} ;filtru de ordinul 1 cu
+frecvența de tăiere la 1kHz
EopAmp Out 0 TABLE{V(IN+,IN-)} (-0.15m,-15) (0.15m,15) ;Amplificator
+operațional cu amplificarea de 100000

```

#### 4. Tipuri de analize:


PSF	.OP
CC	.DC <variabilă> <val_start> <val_final> <val_pas> [înlanțuire] .DC [DEC][OCT] <variabilă><val_start> <val_final> <nr_puncte> [înlanțuire] .DC <variabilă> LIST <valori> [înlanțuire]
Transfer	.TF <variabilă> <sursă_intrare>
Senzitivitate	.SENS <listă_variabilă_de_ieșire>
AC	.AC [LIN] [OCT] [DEC] <număr_puncte> <fstart> <fend>
Zgomot	.NOISE <V(<nod1>[,<nod2>])> <nume_sursă> [valoare_interval]
Tranzitorie	.TRAN <tstep> <tend> [<tstart> <tmax>] [UIC]
Fourier	.FOUR <frecv_fundam> <număr_armmonici> <variabile_ieșire>
Parametrice	.STEP [LIN] <nume_param> <val_start> <val_final> <val_pas> .STEP [DEC] [OCT] <nume_param> <val_start> <val_final> <număr_puncte> .STEP PARAM <nume_param> LIST <valori> .TEMP <valori>
Monte Carlo	.MC <număr_rulări> <tip_analiză> <funcție> [opțiuni]
Cazul cel mai defavorabil	.WCASE <tip_analiză> <variabilă_ieșire> <funcție> [opțiuni]

#### Exemple:

```
.NODESET V(4)=5V ;inițializarea V(4) pentru PSF
.DC Vin -5V 5V 0.25V
.DC DEC Igen 100uA 100mA 5
.DC RES RMOD(R) 1 100 0.5
.DC DEC NPN Qpreamp(IS) 1fA 1nA 5
.DC PARAM RVAL 0 1k 10
.DC TEMP LIST 0 50 100
.DC TEMP 0 100 5
.DC VCE 0 20V 0.2V IB LIST 20uA 40uA 60uA
.TF V(3) Vin
.AC LIN 1000 10Hz 20kHz
.AC DEC 10 1Hz 1MegHz
.AC OCT 5 2H 1.6k
.NOISE V(5) Vin 10
.NOISE V(8,4) Iin
*Zgomot total: V(ONoise), V(INoise)
.TRAN 100us 40ms 10ms 10us UIC
.TRAN 100us 40ms
.FOUR 50 V(2,4) I(RL)
.FOUR 1kHz 15 V(12)
.STEP Vin 0V 10V 0.1V
.STEP DEC Igen 1mA 10A 10
.STEP PARAM Rcol LIST 1k 5k 10k
.STEP RES RMOD(R) 1 100 0.5
.STEP DEC NPN Qpilot(BF) 10 100 20
.TEMP 0 10 20 30
.MC 50 TRAN v(12) YMAX LIST
.MC 50 TRAN IC(Q4) YMAX LIST OUTPUT ALL
```

## 5. Definirea componentelor active:

Dioda	D<nume> <NA> <NC> <NMODEL> [AREA] .MODEL <NMODEL> D(par1=val1 par2=val2 ...)
Bopilar	Q<nume> <NC> <NB> <NE> [<NS>] <NMODEL> [AREA] [OFF] +[IC=VBE,VCE] .MODEL <NMODEL> NPN/PNP/LPNP(par1=val1 par2=val2 ...)
JFET	J<nume> <ND> <NG> <NS> <NMODEL> <AREA> [OFF] +[IC=VDS,VGS] .MODEL <NMODEL> NJF/PJF(par1=val1 par2=val2 ...)
MOS-FET	M<nume> <ND> <NG> <NS> <NB> <NMODEL> [L=<val>] +[W=<val>] [M=<val>] .MODEL <NMODEL> NMOS/PMOS[(par1=val1 par2=val2 ...)]



## 6. Funcții SPICE:

ABS(x)	x
SQRT(x)	$\sqrt{x}$
EXP(x)	$e^x$
LOG(x)	$\ln(x)$
LOG10(x)	$\log(10)$
PWR(x,y)	$ x ^y$
PWSR(x,y)	$ x ^y$ , pentru $x > 0$ și $- x ^y$ , pentru $x < 0$
SIN(x), COS(x), TAN(x), ARCTAN(x)	Funcții trigonometrice
D(x)	Derivata lui x în raport cu mărimea de pe abscisă
S(x)	Integrala lui x pe domeniul variabilei de pe abscisă
AVG(x)	Media lui x
RMS(x)	Valoarea efectivă a lui x
M(x)	Modulul lui x
DB(x)	Modulul lui x în decibeli

P(x)	Faza lui x în grade
R(x)	Partea reală a lui x
IMG(x)	Partea imaginară a lui x
MIN(x)	Minimumul părții reale a lui x
MAX(x)	Maximumul părții reale a lui x
G(x)	Întârzierea de grup a lui x exprimată în secunde

## **7. Organizare:**

Tot ceea ce se va lucra în PSPICE va fi salvat într-un director de lucru (C:\Ap\_Lab), iar fișierele se vor denumi după un cod unic **abcdefgh**, unde:

a	Litera secției
b	Numărul stației de lucru
c	Cifra discriminantă a grupei
d	Litera ce desemnează semigrupa
efgh	La alegere

Exemplu: **M37AL1F2** înseamnă: microelectronică, stația 3, grupa 5407, semigrupa A, laborator 1, fișierul al doilea.

## **8. Considerații PSpice Msim 5.2:**

Dacă anumite componente (part) nu se găsesc în nici una din bibliotecile disponibile, se alege una din soluțiile:

1. Se creează modelele și simbolurile grafice necesare;
2. Mai simplu, nu se mai lucrează în mod grafic (cu Schematics), ci cu ajutorul unui editor de texte (Notepad) se creează direct fișierul text (cu extensia .cir) corespunzător circuitului, care ulterior va fi încărcat în PSPICE; astfel se renunță la crearea fișierului schemă (cu extensia .sch).
3. În Schematics se preia un simbol grafic al unei componente cu aceeași funcționalitate ca și componenta dorită. Se modifică apoi numele de model. Se creează un fișier bibliotecă (cu extensia .lib) care să conțină descrierea modelului respectiv. Se specifică apoi calea spre biblioteca respectivă, prin introducerea în schemă a unei componente speciale, numite LIB.

